
	Reason for change:
	[bookmark: OLE_LINK1]3GPP TS 29.500 has defined the 3gpp-Sbi-Consumer-Info header, which allows the NF consumer to explicitly indicate the supported API version, supported features, accepting encoding as well as the callback prefix/roots (inter-PLMN/intra-PLMN) of the notification callback URIs. This header is usually provided in a subscription creation by NF consumer to NF producer to clarify the NF consumer characteristics and the producer will store the header for future use (i.e. when generating and delivering of the notifications).

When the subscription is done by an intermediate NF on behalf of the source NF, e.g. the UDM subscribes to the AMF events on behalf of the NEF, both the source NF and the intermediate NF may need to provide the NF consumer information with this header. These 3gpp-Sbi-Consumer-Infos will be further populated from the source AMF to the target AMF upon AMF change procedure. 

[bookmark: _Toc89035177][bookmark: _Toc89064975][bookmark: _Toc89180274][bookmark: _Toc97071953][bookmark: _Toc120051355][bookmark: _Toc161843435]6.1.6.2.66            Type: AmfEventSubscriptionAddInfo
Table 6.1.6.2.66-1: Definition of type AmfEventSubscriptionAddInfo
	Attribute name
	Data type
	P
	Cardinality
	Description

	Skipped for Clarity

	nfConsumerInfo
	array(string)
	C
	1..N
	When present, this IE shall contain the NF Service Consumer information received together with the AMF event subscription and entries of the array shall be set to the value of the 3gpp-Sbi-Consumer-Info header defined in clause 5.2.3.3.7 of 3GPP TS 29.500 [4], without the header name.



However, in current definition, it is not specified which header is associated with which NF consumer (intermediate NF or source NF), it would be problematic for the target AMF to determine which 3gpp-Sbi-Consumer-Info header it shall use to send notifications. 

This CR propose to add an indication of the intermediate NF, to allow the NF producer to clearly distinguish the owner.

Rev1: At the last CT4 meeting, there was a proposal to include NF ID in the 3gpp-Sbi-Consumer-Info, the NF ID is a Mandatory IE present in the Subscription Creation request, however, when that information (3gpp-Sbi-Consumer-Infos) is transferred to the target AMF subsequently, the target AMF has no idea which NF ID is corresponding to the UDM, and which is for NEF. It is not so good to use the presence of NF ID to indicate that it is the 3gpp-Sbi-Consumer-Info for an intermediate NF, protocol wise, since the NF ID may be used for other purpose.

So, it is better to introduce an indication to indicate if it is an intermediate NF.


	
	

	Summary of change:

2/ Describe that intermediate NF may provide its own consumer info header in 6.2.2

3/ Update ABNF accordingly.


	
	

	Consequences if not approved:

* * * First Change * * * *
[bookmark: _Toc160973231]5.2.3.3.7	3gpp-Sbi-Consumer-Info
This header contains a comma-delimited list of NF service consumer information from an HTTP client (as NF service consumer).
The encoding of the header follows the ABNF as defined in IETF RFC 9110 [11].

Sbi-Consumer-Info-Header = "3gpp-Sbi-Consumer-Info:" OWS 1#( supportedService ";" OWS supportedVersions [ ";" OWS supportedFeatures ] [ ";" OWS acceptEncoding ] [ ";" OWS callback-uri-prefix ] [ ";" OWS intraPlmnCallbackRoot ";" OWS interPlmnCallbackRoot ] [ ";" OWS intermediateNfIndication ] ) OWS

supportedService = "service=" servicename

servicename = 1*( "-" / %x30-39 / %x41-5A / "_" / %x61-7A )

supportedVersions = "apiversion=" "(" OWS [ apimajorversion *( RWS apimajorversion ) OWS ] ")"

apimajorversion = %x31-39 [ *DIGIT ]

supportedFeatures = "supportedfeatures=" features

features = *HEXDIG

acceptEncoding = "acceptencoding=" %x22 encodingList %x22

encodingList = #( codings [ weight ] )

intraPlmnCallbackRoot = "intraPlmnCallbackRoot=" DQUOTE sbi-scheme "://" sbi-authority
                        [ prefix ] DQUOTE

interPlmnCallbackRoot = "interPlmnCallbackRoot=" DQUOTE sbi-scheme "://" sbi-authority
                        [ prefix ] DQUOTE

sbi-scheme = "http" / "https"

sbi-authority = host [ ":" port ]

port = *DIGIT

prefix = path-absolute ; path-absolute production rule from IETF RFC 3986, clause 3.3

callback-uri-prefix = "callback-uri-prefix=" DQUOTE prefix DQUOTE

intermediateNfIndication = "intermediate-nf=" intermeidateNfValue

intermeidateNfValue      = "true"


"service" (Mandatory parameter): Supported Service parameter indicates the name of a service, as defined in 3GPP TS 29.510 [8], which is supported by the sender as NF service consumer.
"apiversion" (Mandatory parameter): Supported Versions parameter indicates the major version(s) of the service API that are supported by the sender as NF service consumer.
"supportedfeatures" (Optional parameter): Supported Features parameter carries a string containing a bitmask in hexadecimal representation, as specified for SupportedFeatures data type in 3GPP TS 29.571 [13], to indicate the feature(s) of the service API that are supported by the sender as NF service consumer.
"acceptencoding" (Optional parameter): Accept Encoding carries a string indicating the accepted content encodings supported by the sender as NF service consumer, when receiving notifications defined by the service. In the ABNF definition, "codings" and "weight" are defined in IETF RFC 9110 [11] clauses 12.5.3 and 12.4.2.
"intraPlmnCallbackRoot", "interPlmnCallbackRoot" (Optional parameters): intra plmn callback root and inter plmn callback root supported by the sender as NF service consumer, for the indicated service.
"callback-uri-prefix" (Optional parameter): The NF service consumer may include this parameter when providing a Callback URI when the authority part of the Callback URI is shared by several NF service consumer instances. The NF service consumer may also include this header when providing a Callback URI including a prefix, for use during NF service consumer reselection, when binding procedures are not supported. When present, the "callback-uri-prefix" shall be a path-absolute as specified IETF RFC 3986 [14] (i.e. the first path segment(s) after the authority) which is part of the Callback URI provided by a NF service consumer in the corresponding service request message sent to a NF service producer. The authority and "callback-uri-prefix" in the Callback URI shall uniquely identify a consumer service instance. See clause 6.12.1 for the usage of this parameter.
"intermediate-nf" (Optional parameter): The intermediate NF indication parameter is a boolean set to true indicating that the header is provided by an intermediate NF as the NF service consumer.
EXAMPLE 1:	The NF consumer supports Namf_EventExposure OpenAPI "v1" without any optional feature:
	3gpp-Sbi-Consumer-Info: service=namf-evts; apiversion=(1)
EXAMPLE 2:	The NF consumer supports Nsmf_EventExposure OpenAPI "v1" and "v2" with optional feature number 1 and accepted encoding "gzip":
	3gpp-Sbi-Consumer-Info: service=nsmf-event-exposure; apiversion=(1 2); supportedfeatures=01; acceptencoding="gzip; q=1.0, *;q=0.5"
EXAMPLE 3:	The NF consumer supports both Namf_EventExposure OpenAPI "v1" and Nsmf_EventExposure OpenAPI "v2":
	3gpp-Sbi-Consumer-Info: service=namf-evts; apiversion=(1), service=nsmf-event-exposure; apiversion=(2)
[bookmark: _PERM_MCCTEMPBM_CRPT57490061___5]EXAMPLE 4:	An AMF service instance supports Nsmf_PDUSession OpenAPI "v1", provides the callback URI https://amf45.operator.com/servinst123/pdusession, whereby the authority is shared by more than one AMF service instance, while the prefix "/servinst123" uniquely identifies a specific AMF service instance:

3gpp-Sbi-Consumer-Info: service=nsmf-pdusession; apiversion=(1); callback-uri-prefix="/servinst123"
[bookmark: _PERM_MCCTEMPBM_CRPT57490062___5]EXAMPLE 5:	The NF consumer supports Namf_EventExposure OpenAPI "v1" and sends intra PLMN callback root "https://operator.com" and inter PLMN callback root "https://5gc.mnc012.mcc345.3gppnetwork.org" in the header:
	3gpp-Sbi-Consumer-Info: service=namf-evts; apiversion=(1); intraPlmnCallbackRoot="https://operator.com"; interPlmnCallbackRoot="https://5gc.mnc012.mcc345.3gppnetwork.org"
EXAMPLE 6:	The intermediate NF service consumer (e.g. UDM) Namf_EventExposure OpenAPI "v1" and sends intra PLMN callback root "https://intermediate-nf-authority.com" and inter PLMN callback root "https://5gc.mnc012.mcc345.3gppnetwork.org" in the header:
	3gpp-Sbi-Consumer-Info: service=namf-evts; apiversion=(1); intraPlmnCallbackRoot="https://intermediate-nf-authority.com"; interPlmnCallbackRoot="https://5gc.mnc012.mcc345.3gppnetwork.org";intermediate-nf=true

* * * Next Change * * * *
[bookmark: _Toc160973273]6.2.2	Subscription on behalf of NF Service Consumer
When an NF service consumer subscribes to an intermediate NF for events which may be detected and reported directly by target NF (e.g. an NEF subscribes to Location Reporting event at AMF via UDM and AMF directly reports to the NEF), the NF service consumer may include the "3gpp-Sbi-Consumer-Info" header in the subscription creation request to indicate the supported API versions, features and accepted encodings of the service on the target NF.
When subscribing to the target NF and requiring the target NF to directly report to NF service consumer, the intermediate NF shall invoke the highest API version at the target NF which is supported by the NF service consumer (if indicated) and the intermediate NF. The intermediate NF shall include all received "3gpp-Sbi-Consumer-Info" header(s) in the subscription creation request sent to the target NF.
If the target NF received this header in event subscription creation, the target NF shall generate the request body according to the supported feature(s) and accepted encodings of the NF service consumer for notifications directly sent to the NF service consumer.
Based on operator policy, the NF service consumer may provide a default inter-PLMN or intra-PLMN callback URI in a subscription request to the intermediate NF.
The NF Service Consumer may also include, for each provided service, the following information in the "3gpp-Sbi-Consumer-Info" header(s):
-	the intraPlmnCallbackRoot parameter containing the callback root for receiving intra-PLMN notifications, and
-	the interPlmnCallbackRoot parameter containing the callback root for receiving inter-PLMN notifications.
Upon receiving a subscription request including the above information in the "3gpp-Sbi-Consumer-Info" header, the intermediate NF should check if the target NF is in VPLMN or HPLMN and adapt the callback Root of the callback URI according to the information received from the NF service consumer. For instance, if the NF service consumer included an inter-PLMN callback URI in the subscription request:
-	if the target NF is in HPLMN, then the intermediate NF should replace the callback Root of the callback URI in the subscription request with the provided intraPlmnCallbackRoot while sending the subscription creation request to the target NF; and
-	if the target NF is in VPLMN, then the intermediate NF shall not change the notification/callback URI.
In either case, the Intermediate NF should then forward the "3gpp-Sbi-Consumer-Info" header in the subscription creation request sent to the target NF.
If the intermediate NF expects to receive notifications from the target NF (e.g. the UDM may provide the callback URI to receive subscription change event notifications from the AMF when subscribing to AMF events on behalf of the NEF), to enable the target NF to use, e.g., the proper callback URI, supportedFeature, the intermediate NF may provide the "3gpp-Sbi-Consumer-Info" header for itself in the subscription creation request to the target NF. The intermediate NF should include the intermediate NF indication parameter in the header.
When the target NF is an AMF, the source AMF should forward the information in the received "3gpp-Sbi-Consumer-Info" header(s) to the target AMF during inter-AMF mobility. If the target AMF received intraPlmnCallbackRoot and interPlmnCallbackRoot parameters in the "3gpp-Sbi-Consumer-Info" header information from the source AMF, the target AMF should determine the PLMN of the NF Service Consumer and adapt the callback root of the callback URI correspondingly.

* * * Next Change * * * *
[bookmark: _Toc160973423]D.2	ABNF definitions (Filename: "TS29500_CustomHeaders.abnf")

; ----------------------------------------
;   RFC 5234
; ----------------------------------------

HTAB   = %x09 ; horizontal tab

LF     = %x0A ; linefeed

CR     = %x0D ; carriage return

SP     = %x20

DQUOTE = %x22 ; " (Double Quote)

DIGIT  = %x30-39 ; 0-9

ALPHA  = %x41-5A / %x61-7A ; A-Z / a-z

VCHAR  = %x21-7E ; visible (printing) characters

WSP    = SP / HTAB ; white space

CRLF   = CR LF ; Internet standard newline

HEXDIG = DIGIT / "A" / "B" / "C" / "D" / "E" / "F"



; ----------------------------------------
;   RFC 3986
; ----------------------------------------

unreserved    = ALPHA / DIGIT / "-" / "." / "_" / "~"

pct-encoded   = "%" HEXDIG HEXDIG

sub-delims    = "!" / "$" / "&" / "'" / "(" / ")" / "*" / "+" / "," / ";" / "="

pchar         = unreserved / pct-encoded / sub-delims / ":" / "@"

segment       = *pchar

segment-nz    = 1*pchar

path-abempty  = *( "/" segment )

path-absolute = "/" [ segment-nz *( "/" segment ) ]

path-rootless = segment-nz *( "/" segment )

path-empty    = 0pchar

IPvFuture     = "v" 1*HEXDIG "." 1*( unreserved / sub-delims / ":" )

dec-octet     = "25" %x30-35 / "2" %x30-34 DIGIT / "1" 2DIGIT / %x31-39 DIGIT / DIGIT

h16           = 1*4HEXDIG

ls32          = ( h16 ":" h16 ) / Ipv4address

Ipv4address   = dec-octet "." dec-octet "." dec-octet "." dec-octet

Ipv6address   = 6( h16 ":" ) ls32 
              / "::" 5( h16 ":" ) ls32 
              / [ h16 ] "::" 4( h16 ":" ) ls32 
              / [ *1( h16 ":" ) h16 ] "::" 3( h16 ":" ) ls32 
              / [ *2( h16 ":" ) h16 ] "::" 2( h16 ":" ) ls32
              / [ *3( h16 ":" ) h16 ] "::" h16 ":" ls32
              / [ *4( h16 ":" ) h16 ] "::" ls32
              / [ *5( h16 ":" ) h16 ] "::" h16
              / [ *6( h16 ":" ) h16 ] "::"

IP-literal    = "[" ( Ipv6address / IpvFuture  ) "]"

reg-name      = *( unreserved / pct-encoded / sub-delims )

host          = IP-literal / Ipv4address / reg-name

port          = *DIGIT

scheme        = ALPHA *( ALPHA / DIGIT / "+" / "-" / "." )

userinfo      = *( unreserved / pct-encoded / sub-delims / ":" )

authority     = [ userinfo "@" ] host [ ":" port ]

hier-part     = "//" authority path-abempty / path-absolute / path-rootless / path-empty

query         = *( pchar / "/" / "?" )

fragment      = *( pchar / "/" / "?" )

URI           = scheme ":" hier-part [ "?" query ] [ "#" fragment ]



; ----------------------------------------
;   RFC 5322
; ----------------------------------------

obs-FWS         = 1*WSP *( CRLF 1*WSP )

FWS             = ( [ *WSP CRLF ] 1*WSP ) / obs-FWS

obs-NO-WS-CTL   = %d1-8 / %d11 / %d12 / %d14-31 / %d127

obs-ctext       = obs-NO-WS-CTL

ctext           = %d33-39 / %d42-91 / %d93-126 / obs-ctext

obs-qp          = "\" ( %d0 / obs-NO-WS-CTL / LF / CR )

quoted-pair     = ( "\" ( VCHAR / WSP ) ) / obs-qp

ccontent        = ctext / quoted-pair / comment

comment         = "(" *( [ FWS ] ccontent ) [ FWS ] ")"

CFWS            = ( 1*( [ FWS ] comment ) [ FWS ] ) / FWS

day-name        = "Mon" / "Tue" / "Wed" / "Thu" / "Fri" / "Sat" / "Sun"

obs-day-of-week = [ CFWS ] day-name [ CFWS ]

day-of-week     = ( [ FWS ] day-name ) / obs-day-of-week

obs-day         = [ CFWS ] 1*2DIGIT [ CFWS ]

day             = ( [ FWS ] 1*2DIGIT FWS ) / obs-day

month           = "Jan" / "Feb" / "Mar" / "Apr" / "May" / "Jun"
                / "Jul" / "Aug" / "Sep" / "Oct" / "Nov" / "Dec"

obs-year        = [ CFWS ] 2*DIGIT [ CFWS ]

year            = ( FWS 4*DIGIT FWS ) / obs-year

date            = day month year

obs-hour        = [ CFWS ] 2DIGIT [ CFWS ]

hour            = obs-hour / 2DIGIT

obs-minute      = [ CFWS ] 2DIGIT [ CFWS ]

minute          = obs-minute / 2DIGIT

obs-second      = [ CFWS ] 2DIGIT [ CFWS ]

second          = obs-second / 2DIGIT

time-of-day     = hour ":" minute [ ":" second ]

obs-zone        = "UT" / "GMT" / "EST" / "EDT" / "CST" / "CDT" / "MST" / "MDT"
                / "PST" / "PDT" / %d65-73 / %d75-90 / %d97-105 / %d107-122

zone            = ( FWS ( "+" / "-" ) 4DIGIT ) / obs-zone

time            = time-of-day zone

date-time       = [ day-of-week "," ] date time [ CFWS ]



; ----------------------------------------
;   RFC 6749
; ----------------------------------------

NQCHAR = %x21 / %x23-5B / %x5D-7E



; ----------------------------------------
;   RFC 9110
; ----------------------------------------

OWS            = *( SP / HTAB )

RWS            = 1*( SP / HTAB )

tchar          = "!" / "#" / "$" / "%" / "&" / "'" / "*" / "+" / "-"
               / "." / "^" / "_" / "`" / "|" / "~" / DIGIT / ALPHA

token          = 1*tchar

BWS            = OWS

quoted-string  = DQUOTE *( qdtext / quoted-pair ) DQUOTE

qdtext         = HTAB / SP / %x21 / %x23-5B / %x5D-7E / obs-text

obs-text       = %x80-FF

date1          = day-rfc9110 SP month-rfc9110 SP year-rfc9110 ; e.g., 02 Jun 1982

day-rfc9110    = 2DIGIT

month-rfc9110  = %x4A.61.6E ; "Jan", case-sensitive
               / %x46.65.62 ; "Feb", case-sensitive
               / %x4D.61.72 ; "Mar", case-sensitive
               / %x41.70.72 ; "Apr", case-sensitive
               / %x4D.61.79 ; "May", case-sensitive
               / %x4A.75.6E ; "Jun", case-sensitive
               / %x4A.75.6C ; "Jul", case-sensitive
               / %x41.75.67 ; "Aug", case-sensitive
               / %x53.65.70 ; "Sep", case-sensitive
               / %x4F.63.74 ; "Oct", case-sensitive
               / %x4E.6F.76 ; "Nov", case-sensitive
               / %x44.65.63 ; "Dec", case-sensitive

year-rfc9110   = 4DIGIT

codings        = content-coding / "identity" / "*"

content-coding = token

weight         = OWS ";" OWS "q=" qvalue

qvalue         = ( "0" [ "." *3DIGIT ] ) / ( "1" [ "." *3"0" ] )

credentials    = auth-scheme [ 1*SP ( token68
               / [ ( "," / auth-param ) *( OWS "," [ OWS auth-param ] ) ] ) ]

auth-scheme    = token

auth-param     = token BWS "=" BWS ( token / quoted-string )

token68        = 1*( ALPHA / DIGIT / "-" / "." / "_" / "~" / "+" / "/" ) *"="



; ----------------------------------------
;   3GPP TS 29.500
;
;   Version: 18.5.0 (March 2024)
;
;   (c) 2023, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TSDSI, TTA, TTC).
; ----------------------------------------

;
; Header: 3gpp-Sbi-Message-Priority
;

Sbi-Message-Priority-Header = "3gpp-Sbi-Message-Priority:" OWS
                              ( "3" %x30-31 / %x31-32 DIGIT / DIGIT ) OWS



;
; Header: 3gpp-Sbi-Callback
;

Sbi-Callback-Header = "3gpp-Sbi-Callback:" OWS cbtype *1( ";" OWS "apiversion=" majorversion ) OWS

cbchar              = "-" / "_" / DIGIT / ALPHA

cbtype              = 1*cbchar

majorversion        = *DIGIT  



;
; Header: 3gpp-Sbi-Target-apiRoot
;

Sbi-Target-ApiRoot-Header = "3gpp-Sbi-Target-apiRoot:" OWS sbi-scheme "://" sbi-authority
                            [ prefix ] OWS

sbi-scheme                = "https" / "http"

sbi-authority             = host [ ":" port ]

prefix                    = path-absolute



;
; Header: 3gpp-Sbi-Routing-Binding
;

Sbi-Routing-Binding-Header = "3gpp-Sbi-Routing-Binding:" OWS "bl=" blvalue
                             1*( ";" OWS parameter ) 
                             [ ";" OWS callback-uri-prefix ] OWS

blvalue                    = "nf-instance"
                           / "nf-set"
                           / "nfservice-instance"
                           / "nfservice-set"

parametername              = "nfinst"
                           / "nfset"
                           / "nfservinst"
                           / "nfserviceset"
                           / "servname"
                           / "backupamfinst"
                           / "backupnf"

parameter                  = parametername "=" token



;
; Header: 3gpp-Sbi-Binding
;

Sbi-Binding-Header = "3gpp-Sbi-Binding:" OWS binding-element *( OWS "," OWS binding-element ) OWS

binding-element    = "bl=" blvalue 1*( ";" OWS bh-parameter )
                     [ ";" OWS recoverytime ]
                     [ ";" OWS notif-receiver ]
                     [ ";" OWS "group=" groupvalue ]
                     [ 1*( ";" OWS groupparameter ) ]
                     [ ";" OWS "no-redundancy=" no-red-value ]
                     [ ";" OWS callback-uri-prefix ] OWS

bh-parametername   = parametername / "scope"

bh-parameter       = bh-parametername "=" token

recoverytime       = "recoverytime=" OWS DQUOTE date-time DQUOTE

notif-receiver     = "nr=" URI

groupvalue         = "true" / "false"

groupparametername = "oldgroupid"
                   / "groupid"
                   / "uribase"
                   / "oldnfinst"
                   / "oldservset"
                   / "oldservinst"
                   / "guami"

groupparameter     = groupparametername "=" token

no-red-value       = "true" 



;
; Header: 3gpp-Sbi-Producer-Id
;

Sbi-Producer-Id-Header = "3gpp-Sbi-Producer-Id:" OWS "nfinst=" nfinst
                         [ OWS ";" OWS "nfservinst=" nfservinst ]
                         [ OWS ";" OWS "nfset=" nfset ]
                         [ OWS ";" OWS "nfserviceset=" nfserviceset ] OWS

nfinst                 = 8HEXDIG "-" 4HEXDIG "-" 4HEXDIG "-" 4HEXDIG "-" 12HEXDIG

nfservinst             = token

nfset                  = token

nfserviceset           = token



;
; Header: 3gpp-Sbi-Oci
;

Sbi-Oci-Header  = "3gpp-Sbi-Oci:" OWS oci-element *( OWS "," OWS oci-element ) OWS

oci-element     = timestamp ";" RWS validityPeriod ";" RWS olcMetric ";" RWS olcScope

timestamp       = "Timestamp:" RWS DQUOTE date-time DQUOTE

validityPeriod  = "Period-of-Validity:" RWS 1*DIGIT "s"

olcMetric       = "Overload-Reduction-Metric:" RWS ( "100" / %x31-39 DIGIT / DIGIT ) "%"

olcScope        = nfProducerScope / nfConsumerScope / scpScope / seppScope

nfProducerScope = ( ( "NF-Instance:" RWS nfinst ) 
                  / ( "NF-Set:" RWS nfset ) 
                  / ( "NF-Service-Instance:" RWS nfservinst [ ";" RWS "NF-Inst:" RWS nfinst ] )
                  / ( "NF-Service-Set:" RWS nfserviceset )
                  ) [ ";" RWS sNssaiList ";" RWS dnnList ]

nfConsumerScope = ( "NFC-Instance:" RWS nfinst [ ";" RWS "Service-Name:" RWS servname ] )
                / ( "NFC-Set:" RWS nfset [ ";" RWS "Service-Name:" RWS servname ] )
                / ( "NFC-Service-Instance:" RWS nfservinst [";" RWS "NF-Inst:" RWS nfinst ] )
                / ( "NFC-Service-Set:" RWS nfserviceset )
                / ( "Callback-Uri:" RWS DQUOTE URI DQUOTE *( RWS "&" RWS DQUOTE URI DQUOTE ) )

servname        = token

scpScope        = "SCP-FQDN:" RWS fqdn

seppScope       = "SEPP-FQDN:" RWS fqdn

fqdn            = token

dnnList         = "DNN:" RWS 1*tchar *( RWS "&" RWS 1*tchar )

sNssaiList      = "S-NSSAI:" RWS snssai *( RWS "&" RWS snssai )

snssai          = 1*tchar



;
; Header: 3gpp-Sbi-Lci
;

Sbi-Lci-Header    = "3gpp-Sbi-Lci:" OWS lc-element *( OWS "," OWS lc-element ) OWS

lc-element        = timestamp ";" RWS lcMetric ";" RWS lcScope

lcMetric          = "Load-Metric:" RWS ( "100" / %x31-39 DIGIT / DIGIT ) "%"

lcScope           = lcNfProducerScope / scpScope / seppScope

lcNfProducerScope = ( ( "NF-Instance:" RWS nfinst )
                    / ( "NF-Set:" RWS nfset )
                    / ( "NF-Service-Instance:" RWS nfservinst [ ";" RWS "NF-Inst:" RWS nfinst ] )
                    / ( "NF-Service-Set:" RWS nfserviceset )
                    ) [ ";" RWS sNssaiList ";" RWS dnnList ";" RWS relativeCapacity ]

relativeCapacity  = "Relative-Capacity:" RWS ( "100" / 1*2DIGIT ) "%"



;
; Header: 3gpp-Sbi-Client-Credentials
;

Sbi-Client-Credentials-Header = "3gpp-Sbi-Client-Credentials:" OWS jwt OWS

jwt                           = 1*b64urlchar "." 1*b64urlchar "." 1*b64urlchar

b64urlchar                    = ALPHA / DIGIT / "-" / "_"



;
; Header: 3gpp-Sbi-Source-NF-Client-Credentials
;

Sbi-Source-NF-Client-Credentials-Header = "3gpp-Sbi-Source-NF-Client-Credentials:" OWS jwt OWS



;
; Header: 3gpp-Sbi-Nrf-Uri
;

Sbi-Nrf-Uri-Header = "3gpp-Sbi-Nrf-Uri:" OWS nrfUriParam *( OWS ";" OWS nrfUriParam ) OWS

nrfUriParam        = nrfUriParamName ":" RWS ( nrfUriParamValue1 / nrfUriParamValue2 )

nrfUriParamName    = "nnrf-disc" / "nnrf-nfm" / "nnrf-oauth2" / "oauth2-requested-services" / token

nrfUriParamValue1  = DQUOTE URI DQUOTE

nrfUriParamValue2  = ( nrfServiceName *( RWS "&" RWS nrfServiceName ) )

nrfServiceName     = "nnrf-disc" / "nnrf-nfm"



;
; Header: 3gpp-Sbi-Target-Nf-Id
;

Sbi-Target-Nf-Id-Header = "3gpp-Sbi-Target-Nf-Id:" OWS "nfinst=" nfinst
                          [ ";" OWS "nfservinst=" nfservinst ] OWS



;
; Header: 3gpp-Sbi-Max-Forward-Hops
;

Sbi-Max-Forward-Hops-Header = "3gpp-Sbi-Max-Forward-Hops:" OWS ( %x31-39 DIGIT / DIGIT )
                              ";" OWS "nodetype=" nodetypevalue OWS

nodetypevalue               = "scp"



;
; Header: 3gpp-Sbi-Originating-Network-Id
;

Sbi-Originating-Network-Id-Header = "3gpp-Sbi-Originating-Network-Id:" OWS 3DIGIT "-" 2*3DIGIT
                                    [ "-" 11HEXDIG ] [ ";" OWS srcinfo ] OWS

srcinfo                           = "src" ":" RWS srctype "-" srcfqdn

srctype                           = "SCP" / "SEPP"

srcfqdn                           = 4*( ALPHA / DIGIT / "-" / "." )



;
; Header: 3gpp-Sbi-Access-Scope
;

Sbi-Access-Scope-Header = "3gpp-Sbi-Access-Scope:" OWS scope-token *( SP scope-token ) OWS

scope-token             = 1*NQCHAR



;
; Header: 3gpp-Sbi-Other-Access-Scopes
;

Sbi-Other-Access-Scopes-Header = "3gpp-Sbi-Other-Access-Scopes:" OWS scope-token
                                 *( SP scope-token ) OWS



;
; Header: 3gpp-Sbi-Access-Token
;

Sbi-Access-Token-Header = "3gpp-Sbi-Access-Token:" OWS credentials OWS



;
; Header: 3gpp-Sbi-Target-Nf-Group-Id
;

Sbi-Target-Nf-Group-Id-Header = "3gpp-Sbi-Target-Nf-Group-Id:" OWS "nfgid=" nfGroupIdValue OWS

nfGroupIdValue                = DQUOTE token DQUOTE



;
; Header: 3gpp-Sbi-Nrf-Uri-Callback
;

Sbi-Nrf-Uri-Callback-Header = "3gpp-Sbi-Nrf-Uri-Callback:" OWS
                              nrfUriCallbackParam *( OWS ";" OWS nrfUriCallbackParam ) OWS

nrfUriCallbackParam         = nrfUriCallbackParamName ":" RWS nrfUriCallbackParamValue

nrfUriCallbackParamName     = "nnrf-disc" / "nnrf-nfm" / token

nrfUriCallbackParamValue    = DQUOTE URI DQUOTE



;
; Header: 3gpp-Sbi-NF-Peer-Info
;

Sbi-NF-Peer-Info-Header = "3gpp-Sbi-NF-Peer-Info:" OWS peerinfo *( ";" OWS peerinfo ) OWS

peerinfo                = peertype "=" token

peertype                = "srcinst"
                        / "srcservinst"
                        / "srcscp"
                        / "srcsepp"
                        / "dstinst"
                        / "dstservinst"
                        / "dstscp"
                        / "dstsepp"



;
; Header: 3gpp-Sbi-Sender-Timestamp
;

Sbi-Sender-Timestamp-Header = "3gpp-Sbi-Sender-Timestamp:" OWS
                              day-name "," SP date1 SP time-of-day "." milliseconds SP "GMT" OWS

milliseconds                = 3DIGIT



;
; Header: 3gpp-Sbi-Max-Rsp-Time
;

Sbi-Max-Rsp-Time-Header = "3gpp-Sbi-Max-Rsp-Time:" OWS 1*5DIGIT OWS



;
; Header: 3gpp-Sbi-Correlation-Info
;

Sbi-Correlation-Info-Header = "3gpp-Sbi-Correlation-Info:" OWS
                              correlationinfo *( ";" OWS correlationinfo ) OWS

correlationinfo             = ctype "-" cvalue

ctype                       = extension-token / "imsi" / "impi" / "suci" / "nai" / "gci" / "gli"
                            / "impu" / "msisdn" / "extid" / "imeisv" / "imei" / "mac" / "eui"

extension-token             = 1*( "!" / "#" / "$" / "%" / "&" / "'" / "*" / "+" / "." / "^" / "_"
                            / "`" / "|" / "~" / DIGIT / ALPHA )

cvalue                      = 1*( tchar / "@" )



;
; Header: 3gpp-Sbi-Alternate-Chf-Id
;

Sbi-Alternate-Chf-Id-Header = "3gpp-Sbi-Alternate-Chf-Id:" OWS
                              "nfinst=" nfinst ";" OWS ( "primary" / "secondary" ) OWS



;
; Header: 3gpp-Sbi-Notif-Accepted-Encoding
;

Sbi-Notif-Accepted-Encoding-Header = "3gpp-Sbi-Notif-Accepted-Encoding:" OWS
                                     encoding-element *( OWS "," OWS encoding-element ) OWS

encoding-element                   = codings [ weight ]



;
; Header: 3gpp-Sbi-Consumer-Info
;

Sbi-Consumer-Info-Header = "3gpp-Sbi-Consumer-Info:" OWS
                           consumer-info-element *( OWS "," OWS consumer-info-element ) OWS

consumer-info-element    = ( supportedService ";" OWS supportedVersions 
                             [ ";" OWS supportedFeatures ]
                             [ ";" OWS acceptEncoding ]
                             [ ";" OWS callback-uri-prefix ]
                             [ ";" OWS intraPlmnCallbackRoot ";" OWS interPlmnCallbackRoot ]
                             [ ";" OWS intermediateNfIndication ]
                           ) [ ";" OWS intraPlmnCallbackRoot ";" OWS interPlmnCallbackRoot ]

supportedService         = "service=" servicename

servicename              = 1*( "-" / %x30-39 / %x41-5A / "_" / %x61-7A )

supportedVersions        = "apiversion=" "(" OWS
                           [ apimajorversion *( RWS apimajorversion ) OWS ] ")"

apimajorversion          = %x31-39 [ *DIGIT ]

supportedFeatures        = "supportedfeatures=" features

features                 = *HEXDIG

acceptEncoding           = "acceptencoding=" %x22 encodingList %x22

encodingList             = [ encoding-element *( OWS "," OWS encoding-element ) ]

intraPlmnCallbackRoot    = "intraPlmnCallbackRoot="
                           DQUOTE sbi-scheme "://" sbi-authority [ prefix ] DQUOTE

interPlmnCallbackRoot    = "interPlmnCallbackRoot="
                           DQUOTE sbi-scheme "://" sbi-authority [ prefix ] DQUOTE

callback-uri-prefix      = "callback-uri-prefix=" DQUOTE prefix DQUOTE

intermediateNfIndication = "intermediate-nf=" intermeidateNfValue

intermeidateNfValue      = "true"


;
; Header: 3gpp-Sbi-Response-Info
;

Sbi-Response-Info-Header = "3gpp-Sbi-Response-Info:" OWS
                           resp-info-param *( OWS ";" OWS resp-info-param ) OWS

resp-info-param          = resp-info-param-name "=" OWS resp-info-param-value

resp-info-param-name     = "request-retransmitted"
                         / "nfinst"
                         / "nfset"
                         / "nfservinst"
                         / "nfserviceset"
                         / "context-transferred"
                         / "no-retry"
                         / token

resp-info-param-value    = token



;
; Header: 3gpp-Sbi-Selection-Info
;

Sbi-Selection-Info-Header = "3gpp-Sbi-Selection-Info:" OWS
                            selection-info-element *( OWS "," OWS selection-info-element ) OWS

selection-info-element    = ( "reselection=" reselectionvalue *( ";" OWS selection-criteria ) )
                          / ( selection-criteria *( ";" OWS selection-criteria ) )

reselectionvalue          = "true" / "false"

selection-criteria        = selection-action "=" token

selection-action          = "not-select-nfservinst"
                          / "not-select-nfserviceset"
                          / "not-select-nfinst"
                          / "not-select-nfset"



;
; Header: 3gpp-Sbi-Interplmn-Purpose
;

Sbi-Interplmn-Purpose-Header = "3gpp-Sbi-Interplmn-Purpose:" OWS N32Purpose ":" OWS
                               additional-info OWS

N32Purpose                   = "ROAMING"
                             / "INTER_PLMN_MOBILITY"
                             / "SMS_INTERCONNECT"
                             / "ROAMING_TEST"
                             / "INTER_PLMN_MOBILITY_TEST"
                             / "SMS_INTERCONNECT_TEST"
                             / "SNPN_INTERCONNECT"
                             / "SNPN_INTERCONNECT_TEST"
                             / "DISASTER_ROAMING"
                             / "DISASTER_ROAMING_TEST"
                             / token

additional-info              = token



;
; Header: 3gpp-Sbi-Request-Info
;

Sbi-Request-Info-Header = "3gpp-Sbi-Request-Info:" OWS req-param *( ";" OWS req-param ) OWS

req-param               = req-param-name "=" OWS req-param-value

req-param-name          = "retrans" 
                        / "redirect"
                        / "reason"
                        / "idempotency-key"
                        / "receivedrejectioncause"
                        / "callback-uri-prefix"
                        / "nfinst"
                        / "nfservinst"
                        / "redirection-cause"
                        / token

req-param-value         = token



;
; Header: 3gpp-Sbi-Retry-Info
;

Sbi-Retry-Info-Header = "3gpp-Sbi-Retry-Info:" OWS retriesindication OWS

retriesindication     = "no-retries"

* * * End of Changes * * * *


